

Discrete Response Models

Vijay Ivaturi, Chris Rackauckas

July 19th, 2019

1 Introduction

In this tutorial we will go over the simulation of discrete responses. Many pharmacometrics scenarios have observables, such as pain scores or counts, which necessarily have to be discrete. Handling this discreteness can be paramount to getting an appropriate data fit and to properly understand the variation.

Luckily, in Pumas, discrete outputs are handled no differently from the rest of the Pumas toolchain. In Pumas, to have a discrete distribution as output, simply have that your derived or observed variables come from a discrete distribution like a `Poisson` process.

1.1 Binary Response Example

First, let's take a look at a binary response. A binary response is a model which gives an output of 0 or 1 with a probability p . In Pumas, this is represented by a `Bernoulli(p)` distribution.

To get started, first let's load `Pumas` and read in some example data:

```
using Pumas, StatsFuns
data = read_pumas(example_nmtran_data("pain_remed"),
    cvs = [:arm, :dose, :conc, :painord, :remed], event_data=false)
```

```
Population
Subjects: 160
Covariates: arm, dose, conc, painord, remed
Observables: dv
```

Next let's implement a model with a binary response. Here, we do not have an `@dynamics` position. Pumas will automatically handle this. In our `derived`, we define a `logistic` from `StatsFuns`

```
import StatsFuns.logistic
binary_model = @model begin
    @param begin
        intercept ∈ RealDomain(init=0.001)
        tvslope ∈ RealDomain(init=0.0001)
        Ω ∈ VectorDomain(1)
    end

    @random begin
```

```

     $\eta \sim \text{MvNormal}(\Omega)$ 
end

@covariates arm dose

@pre begin
    rx = dose > 0 ? 1 : 0
    slope = tvslope*rx
    logit = intercept + slope +  $\eta[1]$ 
end

@derived begin
    pain = logistic(logit)
    dv ~ Bernoulli(logistic(logit))
end
end

PumasModel
Parameters: intercept, tvslope,  $\Omega$ 
Random effects:  $\eta$ 
Covariates: arm, dose
Dynamical variables:
Derived: pain, dv
Observed: pain, dv

```

Note that we could have alternatively defined our @pre like:

```

@pre begin
    logit = intercept + tvslope*(dose > 0 ? 1 : 0) +  $\eta[1]$ 
    logit = intercept + tvslope*Int(dose > 0) +  $\eta[1]$ 
end

```

more directly instead of using logistic.

Now let's fit our model to the data:

```

param = (
    intercept = 0.001,
    tvslope = 0.0001,
     $\Omega$  = [1.0]
)
res = fit(binary_model,data,param,Pumas.LaplaceI())

```

FittedPumasModel

Successful minimization: true

Likelihood approximation: Pumas.LaplaceI
Objective function value: 1083.01
Total number of observation records: 1920
Number of active observation records: 1920
Number of subjects: 160

```

-----
                Estimate
-----
intercept -1.3085
tvslope   1.7386
 $\Omega_1$     1.5374
-----

```

and simulate some outputs:

```
sim = simobs(binary_model,data,res.param)
simdf = DataFrame(sim, include_events=false)
first(simdf,6) # Print only the first 6 rows
```

	id	time	pain	dv	arm	dose	conc	painord	remed
	String	Float64	Float64	Bool	Int64	Int64	Float64	Int64	Int64
1	1	0.0	0.577513	false	2	20	0.0	3	0
2	1	0.5	0.577513	false	2	20	1.15578	1	0
3	1	1.0	0.577513	false	2	20	1.37211	0	0
4	1	1.5	0.577513	false	2	20	1.30058	0	0
5	1	2.0	0.577513	false	2	20	1.19195	1	0
6	1	2.5	0.577513	false	2	20	1.13602	1	0

Note that now our simulation output for `dv` is true/false values pulled with probability given by `logit` dependent on the individual's random effects.

1.2 Poisson Response Example

Next let's use a `Poisson` counting process in our model. Here we generate a population where everyone is receiving the same doses as a covariate.

```
pop = Population(map(i -> Subject(id=i,cvs=(dose = i.*10,),time=[0.0]),1:10))
```

```
Population
  Subjects: 10
  Covariates: dose
```

Now we define our model without dynamics, and directly use the dose information to predict the count for some observable `dv`:

```
poisson_model = @model begin
  @param begin
    tvbase ∈ RealDomain(init=3.0, lower=0.1)
    d50 ∈ RealDomain(init=50, lower=0.1)
    Ω ∈ PSDDomain(fill(0.1, 1, 1))
  end

  @random begin
    η ~ MvNormal(Ω)
  end

  @pre begin
    baseline = tvbase*exp(η[1])
  end

  @covariates dose

  @derived begin
    dv ~ @. Poisson(baseline*(1-dose/(dose + d50)))
  end
end

PumasModel
  Parameters: tvbase, d50, Ω
  Random effects: η
```

```

Covariates: dose
Dynamical variables:
Derived: dv
Observed: dv

```

and simulate runs from the model:

```

sim = simobs(poisson_model, pop)
simdf = DataFrame(sim, include_events=false)

```

	id	time	dv	dose
	String	Float64	Int64	Int64
1	1	0.0	2	10
2	2	0.0	1	20
3	3	0.0	3	30
4	4	0.0	1	40
5	5	0.0	0	50
6	6	0.0	1	60
7	7	0.0	1	70
8	8	0.0	1	80
9	9	0.0	1	90
10	10	0.0	1	100

Here `dv` is an integer output probabilistically dependent on the dose.